Paper 997-2017

Get Tangency Portfolio by SAS/IML ®

Xia Ke Shan, 3GOLDEN Beijing Technologies Co. Ltd., Beijing, China

Peter Eberhardt, Fernwood Consulting Group Inc., Toronto, Canada

Matthew Kastin, I-Behavior Inc., Louisville, Colorado

ABSTRACT

Mean-Variance Model (Modern portfolio theory) maybe the most famous model in financial field. It assesses a portfolio which's the expected return (mean) is maximized under a given risk (variance). It comes from assumption that investor want as high as return while as low as risk as he could when he invested a couple of assets (a portfolio is the collection of many assets).This model could give us the many optimal portfolio (efficient portfolio frontier) when we know every asset's expect return and their covariance matrix. The tangency portfolio is one of such kind of optimal portfolio after we included a riskless rate (government long-term bond rate), which means it has the maximal expect return (mean) and the minimal risk (variance) among all the optimal portfolios. This paper is trying to simulate some dummy data and get this tangency portfolio by IML code.

INTRODUCTION

Suppose we have a portfolio which contains N assets and they are not independent. Each asset's expect return is r_i and standard deviation of the return is σ_i . Due to they are not independent, covariance between any two assets return is COV (r_i , r_j)= σ_{ij} .

Now we need to calculate the mean and covariance of this portfolio:

Suppose the percent of money invest into asset i is X_i , so ΣX_i =1.

The expect return(mean) of this portfolio is : $\Sigma X_i * r_i$

The risk(variance) of this portfolio is : $X*COV*X^T$

(Note: * in variance is matrix multiply operator unlike * in mean, X is row vector which represent X_i, X^T is column vector – the transpose of X, COV is the covariance matrix of the N asset's return)

Take for an example of two assets:

The expect return (mean) of the portfolio is : $X_1 r_1 + X_2 r_2$

The variance of the portfolio is : $X_1^2 \sigma_1^2 + X_2^2 \sigma_2^2 + 2 X_1 X_2 \sigma_{12}$

(σ1 is the standard deviation of asset one, σ2 is the standard deviation of asset two, σ12 is the covariance of asset one and asset two.)

If we generate a lot of portfolios (i.e. a lot of xi , Σxi =1) and get its mean and variance, plot them into a graph, we can get the following picture. There is a frontier curve, which is a hyperbola. Any points on frontier curve represent a best portfolio. These portfolio are what we should take, because for any variance(risk) of portfolio, we can find a point on frontier curve which can get the maximal expect return(mean) of portfolio.

How can we get tangency portfolio? Now we need a riskless rate (government long-term bond rate) r_f which have 0 variance. Tangency portfolio is a point on frontier curve, this point and r_f form a line which is tangency line to frontier curve. As long as we can get the line which has the maximal slope, then we can say we find this tangency portfolio.



EXAMPLE

Suppose we have three assets (oil bond stock), if we already have calculated their expect return and covariance matrix, what percent of one dollar should invest into oil, bond or stock? And we want this portfolio's expect return as maximal, and risk as minimal as it could be. That leads to a tangency line (to maximize the slope) which goes through the riskless rate (government long-term bond rate) and the tangency portfolio which is on the Efficient Portfolio Frontier. It looks like:



Figure 1. Efficient Portfolio Frontier with Three Assets

The following is the IML code to get this tangency portfolio.

```
proc iml;
/* the number of assets - a portfolio contains N assets */
n=3;
/* riskless rate – government long-term bond rate */
f=0.04 ;

/* simulated data – the return of N assets */
asset=j(10000,n,.);
call randseed(1234);
call randgen(asset,'normal',0,0.2);

/* the expect return of N assets */
return=j(1,n,.);
call randgen(return,'uniform');

/* the covariance matrix of N assets */
cov=cov(asset);
```

```
/* portfolio_return is expect return of a porfolio
   portfolio_std is standard deviation of a porfolio  */
/* get the tangency portfolio – maximize slope of line*/
start max_slope(x) global(f,cov,return);
  portfolio_return=sum(return#x);
  portfolio_std=sqrt(x*cov*x`);
/*calculate slope - riskless rate's std is zero */
  k=(portfolio_return-f)/portfolio_std;
return (k);
finish;

/* constraint conditions - x1+x2...=1 and 0<=x1<=1,0<=x2<=1,...*/
con=repeat({0,1,1},1,n)||{. .,. .,0 1};
/* initial value of x1,x2,...*/
x=j(1,n,1/n);
/* maximize object function value*/
optn={1 1};
/* xres is a solution if rc>0 */
call nlpnra(rc,xres,"max_slope",x,optn,con);


/* print the solution if rc>0*/
print rc[l='Return Code'],xres[l='It is a Solution if Return Code > 0'];

/* save the solution for the following proc sgplot */
if rc > 0 then do;
 create want from xres;
 append from xres;
 close;
end;
else print "Can't find a solution";
quit;
```

```
ABSGCONV convergence criterion satisfied.

Return Code
3

It is a Solution if Return Code > 0
0.6293588       0.361734     0.0089071
```

Output 1. Output from IML Code

It is convergent, so we got the tangency portfolio (62.9% should invest into oil, 36.1% should invest into bond, and 0.9% should invest into stock).

The next thing is checking if it is the real tangency portfolio with the same data.

```
/*********************************************************/
/*Now confirm if the solution is right with the same data*/
proc iml;
n=3;
f=0.04 ;
```

```
asset=j(10000,n,.);
call randseed(1234);
call randgen(asset,'normal',0,0.2);

return=j(1,n,.);
call randgen(return,'uniform');

cov=cov(asset);

/* make 10000 random testing portfolios */
reset fuzz;
temp=j(10000,3,.);
call randgen(temp,'exp');
x=temp/temp[,+];

/*get portfolio's expect return and std*/
portfolio=j(nrow(x),2,.);
do i=1 to nrow(x);
 portfolio[i,1]=sum(return#x[i,]);
 portfolio[i,2]=sqrt(x[i,]*cov*x[i,]`);
end;

/*get efficient portfolio frontier*/
indices=cvexhull(portfolio);
hullIndices=indices[loc(indices>0)];
frontier=portfolio[hullIndices,];

/*get tangency portfolio's expect return and std*/
use want;
read all var _num_ into x;
close;
tangency=j(1,2,.);
tangency[1]=sum(return#x);
tangency[2]=sqrt(x*cov*x`);

/*save it for the following proc sgplot*/
create frontier from frontier[c={return std}];
append from frontier;
close;

create tangency from tangency[c={tang_return tang_std}];
append from tangency;
close;

riskless=f||0;
create riskless from riskless[c={riskless_return riskless_std}];
append from riskless;
close;

/*the slope of line between riskless rate and tangency portfolio*/
k=(tangency[1]-riskless[1])/tangency[2];
call symputx('k',k);
quit;

data sgplot;
 set frontier tangency riskless;
run;
```

```
title "Efficient Portfolio Frontier With Three Assets";
proc sgplot data=sgplot;
 lineparm x=riskless_std y=riskless_return slope=&k /
 lineattrs=graphdata6(thickness=4);

 series x=std y=return/ smoothconnect lineattrs=graphdata1(thickness=4)
  name='c' legendlabel='Efficient Portfolio Frontier';

 scatter x=riskless_std y=riskless_return/ name='a'
 legendlabel='Riskless Rate'
 markerattrs=graphdata3(symbol=diamondfilled size=12);

 scatter x=tang_std y=tang_return/ name='b'
 legendlabel='Tangency Portfolio'
 markerattrs=graphdata2(symbol=starfilled size=14);

 keylegend 'a' 'b' 'c' / location=inside position=topleft across=1;
 yaxis label='Expect Return Of Portfolio' grid valuesformat=percent8.2;
 xaxis label='Standard Deviation Of Portfolio' grid;
run;
```

It produces the same diagram at the beginning. It is just the tangency portfolio we are looking for.

REFERENCES

SAS Institution Inc.2014. *SAS/IML® 13.2 User's Guide*. Cary, NC: SAS Institute Inc.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Xia Ke Shan
3GOLDEN Beijing Technologies Co. Ltd.
Beijing, China
Phone: +8613521225927
E-mail: 12135835@qq.com

Matthew Kastin
i-behavior Inc.
2051 Dogwood Street, Suite 220
Louisville, CO 80027
E-mail: fried.egg@VERIZON.NET

Peter Eberhardt
Fernwood Consulting Group Inc.
Toronto, Canada
E-mail: peter@fernwood.ca